

Números Felizes e Sucessões de Smarandache: Digressões com o Maple

Delfim F. M. Torres
delfim@mat.ua.pt

Departamento de Matemática
Universidade de Aveiro
3810-193 Aveiro, Portugal

Resumo

Dando jus à matemática experimental, mostramos como o Maple pode ser usado na investigação matemática de algumas questões actualmente sem resposta na Teoria dos Números. A tese defendida é que os alunos de um curso de Matemática podem facilmente usar o computador como um lugar onde se excita e exercita a imaginação.

1 Introdução

Albert Einstein é conhecido por ter dito que “a imaginação é mais importante que o conhecimento”. Se assim é, porquê esperar pelo mestrado ou doutoramento para começar a enfrentar problemas em aberto? Não é a criatividade prerrogativa dos mais novos? Em [3] mostrei como com muito pouco conhecimento é possível debruçar-mo-nos sobre algumas questões actualmente sem resposta na Teoria de Computação. Aqui, e a propósito do ano 2003 ter sido escolhido pela APM como o ano da *Matemática e Tecnologia*, vou procurar mostrar como o computador e um ambiente moderno de computação algébrica, como seja o Maple, podem ser excelentes auxiliares na abordagem a “quebra-cabeças” que a matemática dos números actualmente nos coloca. A minha escolha do sistema Maple prende-se com o facto de ser este o programa informático actualmente usado na cadeira de *Computadores no Ensino da Matemática*, no Departamento de Matemática da Universidade de Aveiro. Desta maneira os meus alunos serão prova viva de que basta um semestre de “tecnologias na educação matemática”, para nos podermos aventurar por “mares ainda não navegados”. O leitor que queira aprender sobre o Maple poderá começar por consultar o nosso site de *Computadores no Ensino da Matemática*: <http://webct.ua.pt/public/compensmat/index.html>.

2 Números felizes

Seja $n \in \mathbb{N}$ um número natural com representação decimal $n = d_k \dots d_0$, $0 \leq d_i \leq 9$ ($i = 0, \dots, k$), e denotemos por $\sigma(n)$ a soma dos quadrados dos dígitos decimais de n : $\sigma(n) = \sum_{i=0}^k (d_i)^2$. Dizemos que n é um *número feliz* se existir um $r \in \mathbb{N}$ tal que $\underbrace{(\sigma \circ \dots \circ \sigma)}_{r \text{ vezes}}(n) = 1$. Por exemplo, 7 é um número feliz ($r = 5$),

$$\sigma(7) = 49, \sigma(49) = 97, \sigma(97) = 130, \sigma(130) = 10, \sigma(10) = 1;$$

enquanto 2 não:

$$\sigma(2) = 4, \sigma(4) = 16, \sigma(16) = 37, \sigma(37) = 58, \sigma(58) = 89, \\ \sigma(89) = 145, \sigma(145) = 42, \sigma(42) = 20, \sigma(20) = 4 \dots$$

Vamos definir em Maple a função característica Booleana dos números felizes. Começamos por definir a função `digitos` que nos devolve a sequência de dígitos de um dado número n

```
> digitos := n -> seq(iquo(irem(n,10^i),10^(i-1)),i=1..length(n));
> digitos(12345);
```

5, 4, 3, 2, 1

A função σ é agora facilmente construída

```
> sigma := n -> add(i^2,i=digitos(n));
> sigma(24);
```

20

O processo de composição da função σ é obtido usando o operador `@` do Maple:

```
> s := (n,r) -> seq((sigma@@i)(n),i=1..r);
> s(7,5);
```

49, 97, 130, 10, 1

```
> s(2,9);
```

4, 16, 37, 58, 89, 145, 42, 20, 4

Para automatizarmos o processo de decisão se um número é feliz ou não, recorreremos a alguma programação. O seguinte procedimento deve ser claro.

```
> feliz := proc(n)
> local L, v;
> L := {};
> v := sigma(n);
> while (not (member(v,L) or v=1)) do
> L := L union {v};
> v := sigma(v);
> end do;
> if (v = 1) then true else false end if;
> end proc;
```

Podemos agora questionar o sistema Maple acerca da felicidade de um determinado número.

```
> feliz(7);
```

true

```
> feliz(2);
```

false

A lista de todos os números felizes até 100 é dada por

```
> select(feliz,[1..100]);
```

[1, 7, 10, 13, 19, 23, 28, 31, 32, 44, 49, 68, 70, 79, 82, 86, 91, 94, 97, 100]

Concluimos então que existem 20 números felizes de entre os primeiros 100 naturais

```
> nops(select(feliz, [$1..100]));
```

20

Existem 143 números felizes não superiores a 1000; 1442 não superiores a 10000; e 3038 não superiores a 20000:

```
> nops(select(feliz, [$1..1000]));
```

143

```
> nops(select(feliz, [$1..10000]));
```

1442

```
> nops(select(feliz, [$1..20000]));
```

3038

Estas últimas experiências com o Maple permitem-nos formular a seguinte conjectura.

Conjectura 1. Cerca de um sétimo de todos os números são felizes.

Uma questão interessante é estudar números felizes consecutivos. De entre os primeiros 1442 números felizes podemos encontrar 238 pares de números felizes consecutivos (o mais pequeno é o (31, 32));

```
> felizDezMil := select(feliz, [$1..10000]):
> nops(select(i->member(i, felizDezMil) and
               member(i+1, felizDezMil), felizDezMil));
```

238

onze ternos de números felizes consecutivos, o mais pequeno dos quais é o (1880, 1881, 1882);

```
> select(i->member(i, felizDezMil) and
         member(i+1, felizDezMil) and
         member(i+2, felizDezMil), felizDezMil);
```

[1880, 4780, 4870, 7480, 7839, 7840, 8180, 8470, 8739, 8740, 8810]

dois quaternos de números felizes consecutivos, o mais pequeno dos quais é o (7839, 7840, 7841, 7842);

```
> select(i->member(i, felizDezMil) and
         member(i+1, felizDezMil) and
         member(i+2, felizDezMil) and
         member(i+3, felizDezMil), felizDezMil);
```

[7839, 8739]

e nenhuma sequência de cinco números felizes consecutivos.

```
> select(i->member(i, felizDezMil) and
         member(i+1, felizDezMil) and
         member(i+2, felizDezMil) and
         member(i+3, felizDezMil) and
         member(i+4, felizDezMil), felizDezMil);
```

[]

Sabe-se que a primeira sequência de cinco números felizes consecutivos começa com o 44488.

```
feliz(44488) and feliz(44489) and feliz(44490) and  
    feliz(44491) and feliz(44492);
```

true

É também conhecida uma sequência de 7 números felizes consecutivos, que começa com o número 7899999999999959999999996 (*vide* [4]).

3 Sucessões de Smarandache

Dada uma sucessão de inteiros $\{u_n\}$, a correspondente sucessão de Smarandache $\{s_n\}$ é definida por concatenação de inteiros como se segue:

$$s_1 = u_1, s_2 = u_1 u_2, \dots, s_n = u_1 \dots u_n, \dots$$

Estamos interessados na sucessão de Smarandache associada aos números felizes. Os primeiros elementos desta sucessão são:

1, 17, 1710, 171013, 17101319, 1710131923, 171013192328, 17101319232831, ...

Começamos por implementar a concatenação de inteiros em Maple.

```
> conc := (a,b) -> a*10~length(b)+b;  
> conc(12,345);
```

12345

Formando a lista dos números felizes até um certo n , e usando a função `conc` acima definida, a correspondente sucessão de Smarandache é facilmente obtida.

```
> sh := proc(n)  
>   local L, R, i;  
>   L := select(feliz, [$1..n]);  
>   R := array(1..nops(L),L);  
>   for i from 2 by 1 while i <= nops(L) do  
>     R[i]:=conc(R[i-1],L[i]);  
>   end do;  
>   return(R);  
> end proc;
```

Como

```
> select(feliz, [$1..31]);
```

[1, 7, 10, 13, 19, 23, 28, 31]

os primeiros 8 valores da sucessão de Smarandache são então

```
> print(sh(31));
```

[1, 17, 1710, 171013, 17101319, 1710131923, 171013192328, 17101319232831]

Existem muitas questões em aberto associadas à sucessão de Smarandache dos números felizes (*vide* [2]). Umaz dizem respeito à existência de números primos na sucessão; outras à existência de números felizes. Façamos agora alguma investigação a este respeito. Usando o Maple é fácil concluir que de entre os primeiros 143 termos da sucessão de Smarandache dos números felizes, apenas 3 são primos.

```
> primos := select(isprime,sh(1000)):
> nops([seq(primos[i],i=1..143)]);
```

3

Se fizermos `print(primos)` vemos que os três primos são $s_2 = 17$, $s_5 = 17101319$ e s_{43} (s_{43} é um primo com 108 dígitos decimais).

```
> primos[2], primos[5];
17, 17101319
> length(primos[43]);
```

108

Apenas são conhecidos estes números primos na sucessão de Smarandache dos números felizes. Permanece por esclarecer se eles serão ou não em número finito (*vide* [1]).

Existem 31 números felizes de entre os primeiros 143 termos da sucessão de Smarandache dos números felizes:

```
> shFelizes := select(feliz,sh(1000)):
> nops([seq(shFelizes[i],i=1..143)]);
```

31

Recorrendo ao comando `print(shFelizes)` vemos que esses números são o $s_1, s_{11}, s_{14}, s_{30}, s_{31}, s_{35}, s_{48}, s_{52}, s_{58}, s_{62}, s_{67}, s_{69}, s_{71}, s_{76}, s_{77}, s_{78}, s_{82}, s_{83}, s_{85}, s_{98}, s_{104}, s_{108}, s_{110}, s_{114}, s_{115}, s_{117}, s_{118}, s_{119}, s_{122}, s_{139}$ e s_{140} . A título de curiosidade, s_{140} tem 399 dígitos:

```
> length(shFelizes[140]);
```

399

Muito existe por esclarecer relativamente à existência de números felizes consecutivos na sucessão de Smarandache dos números felizes. Olhando para os resultados anteriores vemos que o par mais pequeno de números felizes consecutivos é o (s_{30}, s_{31}) ; enquanto o terno mais pequeno é o (s_{76}, s_{77}, s_{78}) . Quantos termos consecutivos são possíveis? É capaz de encontrar exemplos, digamos, de seis números felizes consecutivos? Estas e outras questões estão em aberto (*vide* [1]). Ferramentas como o Maple são boas auxiliares neste tipo de investigações. Fico à espera de algumas respostas da sua parte.

Referências

- [1] S. S. Gupta, *Smarandache sequence of happy numbers*, Smarandache Notions Journal, Vol. 13, no. 1-3, 2002 (see online version at <http://www.shyamsundergupta.com/shappy.htm>).
- [2] R. K. Guy, *Unsolved problems in number theory*, Second edition, Springer, New York, 1994.
- [3] D. F. M. Torres, *O Computador Matemático de Post*, Boletim da Sociedade Portuguesa de Matemática, N^o 46, Abril de 2002, pp. 81-94.
- [4] D. W. Wilson, Sequence A055629 (Jun 05 2000) in the On-Line Encyclopedia of Integer Sequences, <http://www.research.att.com/~njas/sequences>